
New features
- Create MMS orbit plots (mms_orbit_plot.pro) 
- Create a series of images with line/spectra plots + 2D/1D cuts of distributions (mms_flipbookify.pro) 
- Send MMS data to CDF files and Autoplot 
- MMS wrapper for 2D slices (mms_part_slice2d.pro) 
- Updates to mms_part_getspec 

— Bulk velocity subtraction 
— Subtract FPI distribution error, spin-tone 
— Add B-field / spacecraft RAM direction to phi/theta spectrograms 
— 'multipad' option, with mms_part_getpad (create a PAD for different energy ranges very quickly) 

- mirror_data_dir keyword in mms_init (load data from a local network mirror instead of SDC/SPDF) 
- flatten_spectra (quickly create line plots of spectra at specific times by clicking time varying spectra plots) 

New example scripts
- Poynting flux crib sheet (mms_poynting_flux_crib.pro) 
- Plasma beta crib sheet (mms_plasma_beta_crib.pro) 
- Webinar scripts now in mms/examples/webinars folder 



Getting Started 
Requirements

- Windows, Linux, OS X, or Solaris 
- IDL 8.4+ 
- IDL CDF Library 3.6.3+

Getting Started 
Installing SPEDAS

http://spedas.org/wiki/index.php?title=Downloads_and_Installation

Getting Started 
Organization

spedas_gui: Components of the SPEDAS Graphical User 
Interface (GUI) 

external: Code developed externally, but distributed with 
SPEDAS (CDAWeb, Geopack, etc.) 

projects: Mission specific code 

general: General science analysis tools

Getting Started 
MMS Load Routines

mms_load_fgm Fluxgate Magnetometer
mms_load_scm Search-coil Magnetometer
mms_load_fsm L3 combined FGM+SCM data products
mms_load_mec Ephemeris and Coordinates
mms_load_fpi Fast Plasma Investigation

mms_load_hpca Hot Plasma Composition Analyzer
mms_load_eis Energetic Ion Spectrometer

mms_load_feeps Fly’s Eye Energetic Particle Sensor
mms_load_edp Electric-field Double Probe
mms_load_edi Electron Drift Instrument
mms_load_dsp Digital Signal Processor

mms_load_aspoc Active Spacecraft Potential Control
mms_load_tqf Tetrahedron Quality Factor

mms_load_fast_segments Fast survey intervals
mms_load_brst_segments Burst intervals

See folder: /projects/mms/

Getting Started 
Standard Keywords

• trange 

• probes 

• level 

• data_rate 

• datatype

trange=['2015-10-16', '2015-10-17']

probes=[1, 2, 3, 4]

level='l2'

data_rate='srvy'

datatype=['des-moms', 'dis-moms']

• suffix 

• time_clip 

• no_update 

• spdf 

• tplotnames

suffix='_burst_mode'

/time_clip

/no_update

/spdf

tplotnames=tvarnames

Getting Started 
Standard Keywords

Getting Started 
Standard Keywords

• local_data_dir 

• mirror_data_dir 

• cdf_filenames 

• cdf_version 

• min_version 

• latest_version 

• major_version

local_data_dir='/Users/eric/mydata/'

cdf_filenames=data_file_list

cdf_version='3.0.0'

min_version='3.0.0'

/latest_version

mirror_data_dir='/path/to/network/drive/'

/major_version



Getting Started 
Standard Keywords

For a complete list of keywords and their descriptions, see the 
header for the load routine you’re interested in, e.g., 

; PROCEDURE:
;         mms_load_fpi
;         
; PURPOSE:
;         Load data from the Fast Plasma Investigation (FPI) onboard MMS
; 
; KEYWORDS:
;         trange:       time range of interest [starttime, endtime] with the format 
;                       ['YYYY-MM-DD','YYYY-MM-DD'] or to specify more or less than a day 
;                       ['YYYY-MM-DD/hh:mm:ss','YYYY-MM-DD/hh:mm:ss']
;         probes:       list of probes, valid values for MMS probes are ['1','2','3','4']. 
;                       If no probe is specified the default is probe '3'
;         level:        indicates level of data processing. FPI levels currently include 'l2', 
;                       'l1b', 'sitl', 'ql'. 
;         datatype:     valid datatypes are:
;                         Quicklook: ['des', 'dis'] 
;                         SITL: '' (none; loads both electron and ion data from single CDF)
;                         L1b/L2: ['des-dist', 'dis-dist', 'dis-moms', 'des-moms']
;         data_rate:    instrument data rates for MMS FPI include 'fast', 'brst'. 
;         local_data_dir: local directory to store the CDF files; should be set if
;                       you're on *nix or OSX, the default currently assumes Windows (c:\data\mms\)
;         source:       specifies a different system variable. By default the MMS mission system 
;                       variable is !mms
;         get_support_data: load support data (defined by support_data attribute in the CDF)
;         tplotnames:   returns a list of the names of the tplot variables loaded by the load routine
;         no_color_setup: don't setup graphics configuration; use this keyword when you're 
;                       using  this load routine from a terminal without an X server running
;         time_clip:    clip the data to the requested time range; note that if you do not use 
;                       this keyword you may load a longer time range than requested

Getting Started 
SDC Data Availability

https://lasp.colorado.edu/mms/sdc/public/about/processing/
Browse the status of the data:

Browse the data files:
https://lasp.colorado.edu/mms/sdc/public/data/

https://lasp.colorado.edu/mms/sdc/public/search/
Search for data files:

https://lasp.colorado.edu/mms/sdc/public/quicklook/
Browse the quicklook plots at the SDC:

Getting Started 
Simple Example

; load the MMS1 FGM data for October 16, 2015
mms_load_fgm, probe=1, data_rate='srvy', trange=['2015-10-16', '2015-10-17']

; list the tplot variables loaded
tplot_names

; get the data out of a tplot variable
get_data, 'mms1_fgm_b_gsm_srvy_l2_bvec', data=bgsm_vec, dlimits=bgsm_metadata

; get basic info on the IDL vars that hold the B-field data
help, bgsm_vec ; structure, x: times, y: data (x, y, z)
help, bgsm_metadata ; plotting and file metadata

; store the data into a different tplot variable
store_data, 'new_var_with_b_gsm', data=bgsm_vec, dlimits=bgsm_metadata

; plot the newly created variable
tplot, 'new_var_with_b_gsm'

Loading and Plotting Data 
FGM

; load 10 minutes of burst-mode FGM data for probe 1 on October 16, 2015
mms_load_fgm, probe=1, data_rate='brst', trange=['2015-10-16/13:00', '2015-10-16/13:10'], /time_clip

; remove gaps
tdegap, ['mms1_fgm_b_gsm_brst_l2_bvec', 'mms1_fgm_b_gsm_brst_l2_btot'], /overwrite

; plot the vector and magnitude
tplot, ['mms1_fgm_b_gsm_brst_l2_bvec', 'mms1_fgm_b_gsm_brst_l2_btot']

; save the plot to a postscript file
tprint, 'mms1_fgm_b_gsm_brst_l2', /landscape

https://lasp.colorado.edu/mms/sdc/public/datasets/fields/

Loading and Plotting Data 
FGM

Loading and Plotting Data 
FGM

; add burst segments bar to the top of the figure
mms_load_brst_segments, trange=['2015-10-16/13:00', '2015-10-16/13:10']

tplot, 'mms_bss_burst', /add

; save the plot as a postscript file
tprint, 'mms1_fgm_b_gsm_brst_l2_with_burstbar', /landscape

Loading and Plotting Data 
FGM

Loading and Plotting Data 
SCM

; load 10 minutes of burst-mode SCM data for probe 2 on October 16, 2015
mms_load_scm, probe='2', data_rate='brst', trange=['2015-10-16/13:00', '2015-10-16/13:10'], /time_clip

; plot the data
tplot, 'mms2_scm_acb_gse_scb_brst_l2'

; zoom into a few seconds after 13:07
tlimit, ['2015-10-16/13:07:00', '2015-10-16/13:07:02']

; save the plot to a postscript file
tprint, 'mms2_scm_acb_gse_scb_brst_l2', /landscape

https://lasp.colorado.edu/mms/sdc/public/datasets/fields/

Loading and Plotting Data 
SCM

Loading and Plotting Data 
EDP

mms_load_edp, probe=1, level='l2', datatype='dce', trange=['2015-10-16', '2015-10-17'], /latest_version

; plot the E-field data
tplot, 'mms1_edp_dce_gse_fast_l2'

; save the plot to a postscript file
tprint, 'mms1_edp_dce_gse_fast_l2', /landscape

https://lasp.colorado.edu/mms/sdc/public/datasets/fields/

Loading and Plotting Data 
EDP

Loading and Plotting Data 
FPI

; load the FPI electron moments data for October 16, 2015
mms_load_fpi, probe=3, data_rate='fast', trange=['2015-10-16', '2015-10-17'], datatype='des-moms'

; plot the omni-directional electron energy spectra and pitch angle distribution
tplot, ['mms3_des_energyspectr_omni_avg', 'mms3_des_pitchangdist_avg']

; save the plot to a postscript file
tprint, 'mms3_des_spec_and_pad', /landscape

https://lasp.colorado.edu/mms/sdc/public/datasets/fpi/

Loading and Plotting Data 
FPI

Loading and Plotting Data 
FPI

https://lasp.colorado.edu/mms/sdc/public/datasets/fpi/

; we can also plot the FPI density, bulk velocity
; first, we need to combine the velocity into a single tplot variable (Vx, Vy, Vz)
join_vec, ['mms3_des_bulkx_dbcs_fast', 'mms3_des_bulky_dbcs_fast', 'mms3_des_bulkz_dbcs_fast'], 
'mms3_des_bulk_vel_dbcs_fast'

; plot the electron density and flow velocity
tplot, ['mms3_des_bulk_vel_dbcs_fast', 'mms3_des_numberdensity_dbcs_fast']

; save the plot to a postscript file
tprint, 'mms3_des_vel_density', /landscape

Loading and Plotting Data 
FPI

Loading and Plotting Data 
HPCA

; load the HPCA moments data for October 16, 2015
mms_load_hpca, probe=4, data_rate='srvy', trange=['2015-10-16', '2015-10-17'], datatype='moments'

; plot the H+ density, temperature and bulk velocity
tplot, ['mms4_hpca_hplus_number_density', 'mms4_hpca_hplus_scalar_temperature', 
'mms4_hpca_hplus_ion_bulk_velocity']

; save the plot to a postscript file
tprint, 'mms4_hpca_hplus_moments', /landscape

https://lasp.colorado.edu/mms/sdc/public/datasets/hpca/

Loading and Plotting Data 
HPCA

Loading and Plotting Data 
HPCA

https://lasp.colorado.edu/mms/sdc/public/datasets/hpca/

; load the HPCA ion data for October 16, 2015
mms_load_hpca, probe=4, data_rate='srvy', trange=['2015-10-16', '2015-10-17'], datatype='ion'

; calculate the omni-directional flux by averaging over the anodes for the full field of view
mms_hpca_calc_anodes, fov=[0, 360] 
mms_hpca_spin_sum, probe=4

; plot the H+ and O+ spectra
tplot, [‘mms4_hpca_hplus_flux_elev_0-360_spin', $
        ‘mms4_hpca_oplus_flux_elev_0-360_spin']

; save the plot to a postscript file
tprint, 'mms4_hpca_hplus_flux_fullFoV', /landscape

Loading and Plotting Data 
HPCA

Loading and Plotting Data 
EPD EIS

; load the EIS energy x time of flight (ExTOF) ion data for MMS1
mms_load_eis, probe=1, data_rate='srvy', datatype='extof', trange=['2015-10-16', '2015-10-17']

; calculate the EIS pitch angle distribution
mms_eis_pad, probe=1, data_rate='srvy', datatype='extof'

; plot the spin-averaged energy spectra and pitch angle distribution
tplot, ['mms1_epd_eis_extof_proton_flux_omni_spin', 
'mms1_epd_eis_extof_0-1000keV_proton_flux_omni_pad_spin']

; save the plot to a postscript file
tprint, 'mms1_epd_eis_extof_protons', /landscape

https://lasp.colorado.edu/mms/sdc/public/datasets/epd/

Loading and Plotting Data 
EPD EIS

Loading and Plotting Data 
EPD FEEPS

; load the FEEPS electron data for MMS2
mms_load_feeps, probe=2, data_rate='srvy', datatype='electron', trange=['2015-10-16', '2015-10-17']

; calculate the FEEPS pitch angle distribution
mms_feeps_pad, probe=2, data_rate='srvy', datatype='electron'

; plot the spin averaged electron spectra and pitch angle distribution
tplot, ['mms2_epd_feeps_electron_intensity_omni_spin', ‘mms2_epd_feeps_0-1000keV_pad_spin']

; save the plot to a postscript file
tprint, 'mms2_epd_feeps_electrons', /landscape

https://lasp.colorado.edu/mms/sdc/public/datasets/epd/

Loading and Plotting Data 
EPD FEEPS

Loading and Plotting Data 
MEC

; load the ephemeris and coordinates data for October 16, 2015
mms_load_mec, probe=1, trange=['2015-10-16', '2015-10-17']

; plot the spacecraft position and velocity in GSM coordinates
tplot, ['mms1_mec_r_gsm', 'mms1_mec_v_gsm']

Loading and Plotting Data 
Orbits

; create a plot of the spacecraft orbits for December 15, 2015
mms_orbit_plot, probe=[1, 2, 3, 4], trange=['2015-12-15', '2015-12-16']

Loading and Plotting Data 
Tetrahedron Formation

; create the formation plot in GSM coordinates, including the tetrahedron quality factor
mms_mec_formation_plot, '2015-10-16/13:07', coord='gsm', /quality_factor

Loading and Plotting Data 
Send data to Autoplot

mms_load_fgm, trange=['2015-10-16', '2015-10-17'], probe=1
mms_load_feeps, trange=['2015-10-16', '2015-10-17'], probe=1
mms_load_fpi, trange=['2015-10-16', '2015-10-17'], datatype='dis-moms', probe=1

tplot2ap, ['mms1_fgm_b_gse_srvy_l2_bvec', 'mms1_dis_bulkv_gse_fast', 'mms1_dis_numberdensity_fast', 
'mms1_epd_feeps_srvy_l2_electron_intensity_omni_spin']

Analysis Tools
• Coordinate Transformations 

• Minimum Variance Analysis 

• Curlometer Technique 

• Dynamic Power Spectra 

• Spectra from Particle Distributions 

• 2D Particle Slices 

• Flipbook-style figures/movies with 2D Particle Slices 

• Visualizing the Distributions in 3D 

• EIS angle-angle Plots 

• FPI angle-angle Plots

Analysis Tools 
Coordinate Transformations

; load the quaternions from the MEC file
mms_load_mec, probe=3, trange=['2015-10-16/08:00', '2015-10-16/16:00'], /time_clip

; we're going to transform some FGM data
mms_load_fgm, probe=3, trange=['2015-10-16/08:00', '2015-10-16/16:00'], /time_clip

; Cotrans the FGM data from GSM to SM coordinates
mms_qcotrans, 'mms3_fgm_b_gsm_srvy_l2_bvec', 'mms3_fgm_b_sm_srvy_l2_bvec', out_coord='sm'

; plot the data in both GSM and SM coordinates
tplot, ['mms3_fgm_b_gsm_srvy_l2_bvec', 'mms3_fgm_b_sm_srvy_l2_bvec']

; save the plot to a postscript file
tprint, 'mms3_fgm_gsm_sm_coords', /landscape

Analysis Tools 
Coordinate Transformations

Analysis Tools 
Minimum Variance Analysis

; load some FGM data
mms_load_fgm, probe=1, trange=['2015-10-16/08:00', '2015-10-16/16:00'], /time_clip

; create the minimum variance transformation matrix
minvar_matrix_make, 'mms1_fgm_b_gse_srvy_l2_bvec', newname='mva_matrix', $
  tstart='2015-10-16/13:00', tstop='2015-10-16/13:10'

; rotate the B-field 
tvector_rotate, 'mva_matrix', 'mms1_fgm_b_gse_srvy_l2_bvec', newname='mms1_fgm_b_gse_srvy_l2_bvec_mva'

; plot the vector in MVA coordinates
tplot, 'mms1_fgm_b_gse_srvy_l2_bvec_mva'

; save the plot to a postscript file
tprint, 'mms1_fgm_b_gse_srvy_l2_bvec_mva', /landscape

Analysis Tools 
Minimum Variance Analysis

Analysis Tools 
Curlometer Technique

trange = ['2015-10-30/05:15:45', '2015-10-30/05:15:48']

mms_load_fgm, trange=trange, /get_fgm_ephemeris, probes=[1, 2, 3, 4], data_rate='brst'

fields = 'mms'+['1', '2', '3', '4']+'_fgm_b_gse_brst_l2'
positions = 'mms'+['1', '2', '3', '4']+'_fgm_r_gse_brst_l2'

; method #1: mms_curl
mms_curl, trange=trange, fields=fields, positions=positions, suffix='_mms_curl'

tplot, ['divB','curlB','jtotal','jperp','jpar','baryb']+'_mms_curl'

Analysis Tools 
Curlometer Technique

Analysis Tools 
Dynamic Power Spectra

; load some SCM data
mms_load_scm, probe=1, trange=['2015-10-16/00:00', '2015-10-16/04:00']

; calculate the dynamic power spectra
tdpwrspc, 'mms1_scm_acb_gse_scsrvy_srvy_l2', nboxpoints=512,nshiftpoints=512,bin=1

; plot the dynamic power spectra
tplot, ['mms1_scm_acb_gse_scsrvy_srvy_l2_x_dpwrspc', $
        'mms1_scm_acb_gse_scsrvy_srvy_l2_y_dpwrspc', $
        'mms1_scm_acb_gse_scsrvy_srvy_l2_z_dpwrspc']

; save the plot to a postscript file
tprint, 'mms1_scm_acb_gse_scsrvy_l2_dpwrspc', /landscape

Analysis Tools 
Dynamic Power Spectra

mms_part_getspec, probe=1, species='e', trange=['2015-10-16/04:00', '2015-10-16/06:00'], output='energy'

; plot the spectra
tplot, 'mms1_des_dist_fast_energy'

; save the plot to a postscript file
tprint, 'mms1_des_dist_fast_energy', /landscape

Analysis Tools 
Spectra from FPI Distributions

Analysis Tools 
Spectra from FPI Distributions

mms_part_getspec, instrument='hpca', probe=2, species='hplus', trange=['2015-10-16/04:00', 
'2015-10-16/06:00'], output='energy'

tplot, 'mms2_hpca_hplus_phase_space_density_energy'

; save the plot to a postscript file
tprint, 'mms2_hpca_hplus_phase_space_density_energy', /landscape

Analysis Tools 
Spectra from HPCA Distributions

What's New / Plug-in Status
Getting Started
Loading and Plotting Data 
Analysis Tools
Getting Help



Analysis Tools 
Spectra from HPCA Distributions

Analysis Tools 
2D Particle Slices

mms_part_slice2d, data_rate='brst', species='i', probe=1, time='2015-10-16/13:06', export='2dslice', /eps

Analysis Tools 
mms_flipbookify

trange=['2015-10-16/13:06:00', '2015-10-16/13:06:30']
probe=1
data_rate = 'brst'
species = 'i'

mms_load_fgm, trange=trange, probe=probe, /time_clip
mms_load_fpi, trange=trange, probe=probe, datatype='d'+species+'s-moms', /time_clip, data_rate=data_rate

; be sure to make the window large enough in the x-direction for the slices
window, xsize=1000, ysize=650

; store the temperature in the same panel
store_data, 'temp', data='mms1_d'+species+'s_temppara_brst mms1_d'+species+'s_tempperp_brst'

tplot, ['mms1_fgm_b_gse_srvy_l2_bvec', 'mms1_d'+species+'s_heatq_dbcs_brst', 'temp', 
'mms1_d'+species+'s_bulkv_gse_brst', $
  'mms1_d'+species+'s_numberdensity_brst', 'mms1_d'+species+'s_energyspectr_omni_brst']

mms_flipbookify, time_step=10, probe=1, species=species



Analysis Tools 
mms_flipbookify

Analysis Tools 
Visualizing the Distributions in 3D

trange = ['2015-10-20/05:56:30', '2015-10-20/05:56:34']

;get +- 60 seconds of support data (FGM and FPI velocity)
support_trange= time_double(trange) + [-60,60]

; load the FPI electron distribution data
mms_load_fpi, probe=1, trange=trange, data_rate='brst', datatype='des-dist'

; load data into standard structures
dist = mms_get_fpi_dist('mms1_des_dist_brst', trange=trange)

; convert structures to isee_3d data model
data = spd_dist_to_hash(dist)

; load B-field (cyan vector) and velocity (yellow vector) support data
mms_load_fgm, probe=1, trange=support_trange
mms_load_fpi, data_rate='brst', datatype='des-moms', probe=1, trange=support_trange

; combine separate velocity components into a vector
join_vec,  'mms1_des_bulk' + ['x','y','z'] +'_dbcs_brst',  'mms1_des_bulk'

; Once GUI is open select PSD from Units menu
isee_3d, data=data, trange=trange, bfield= 'mms1_fgm_b_gse_srvy_l2_bvec', velocity='mms1_des_bulk'

Analysis Tools 
EIS angle-angle plots

; to plot the angle-angle plots for EIS, use the following tool 
; developed by Ian Cohen and Joe Westlake at JHU/APL
eis_ang_ang, probe='1', trange=['2015-12-15/13:00', '2015-12-15/13:10'], energy_chan=[1, 2, 3, 4]

Analysis Tools 
FPI angle-angle plots

; to plot the angle-angle plots for FPI, use the following tool 
mms_fpi_ang_ang, '2015-10-16/13:06:59.985', /postscript, species='i', data_rate='brst'

Analysis Tools 
FPI angle-angle plots

Examples 
Where to find more examples?

/projects/mms/examples/basic/
Basic examples for each load routine; this is a good place for new users to start

/projects/mms/examples/advanced/
Examples showing more advanced functionality of the plugin

/projects/mms/examples/quicklook/
Scripts that create the QL plots at the SDC; note: these require MMS team member 
access to the SDC to run

/projects/mms/examples/webinars/
Scripts created to showcase plug-in functionality during webinars



Getting Help

• SPEDAS Forum 

• SPEDAS Wiki 

• Ask Eric

https://groups.google.com/forum/#!forum/spedas

http://spedas.org/wiki/

egrimes@igpp.ucla.edu

Getting Help 
Download

Download this presentation:

http://spedas.org/mms/mms_gem_2018.pdf

SPEDAS 
Other Useful Crib Sheets

general/examples/crib_tplot.pro 

general/examples/crib_tplot_annotation.pro 

general/examples/crib_tplot_layout.pro 

general/examples/crib_tplot_range.pro 

general/examples/crib_tplot_ticks.pro 

general/examples/crib_dproc.pro

